Parallel Programming in Erlang (PFP Lecture 9)

John Hughes

What is Erlang?

Erlang ≈ Haskell

- Types
- Lazyness
- Purity
+ Concurrency
+ Syntax

If you know Haskell, Erlang is easy to learn!

QuickSort again

- Haskell

```
qsort [] = []
qsort (x:xs) = qsort [y | y <- xs, y<x]
               ++ [x]
               ++ qsort [y | y <- xs, y>=x]
```

- Erlang

```
qsort([]) -> [];
qsort([X|Xs]) -> qsort([Y || Y <- Xs, Y<X])
               ++ [X]
               ++ qsort([Y || Y <- Xs, Y>=X]).
```

- Haskell

```
qsort [] = []
qsort (x:xs) = qsort [y | y <- xs, y<x]
```

- Erlang

```
qsort([]) -> [];
qsort([X|Xs]) -> qsort([Y || Y <- Xs, Y<X])
                 ++ [X]
                 ++ qsort([Y || Y <- Xs, Y>=X]).
```



QuickSort again

- Haskell

```
qsort [] = []
qsort (x:xs) = q          <- xs
               ++ [x]
               ++ qsort [y | y <- xs
```

- Erlang

```
qsort([]) -> [];
qsort([X|Xs]) -> qsort([Y || Y <- Xs, Y<X])
                 ++ [X]
                 ++ qsort([Y || Y <- Xs, Y>=X]).
```



Qu gain

- Haskell

```
qsort [] = []
qsort (x:xs) = qsort [y | y <- xs, y<x]
               ++ [x]
               ++ [y | y <- xs, y>=x]
```

- Erlang

```
qsort([]) -> [];
qsort([X|Xs]) -> qsort([Y || Y <- Xs, Y<X])
                 ++ [X]
                 ++ qsort([Y || Y <- Xs, Y>=X]).
```

QuickSort again

- Haskell

```
qsort [] = []
qsort (x:xs) = qsort [y | y <- xs, y<x]
               ++ [x]
               ++ qsort [y | y <- xs
```

- Erlang

```
qsort([]) -> [];
qsort([X|Xs]) -> qsort([Y || Y <- Xs, Y<X])
               ++ [X]
               ++ qsort([Y || Y <- Xs, Y>=X]).
```



foo.erl

Declare the module name

```
-module(foo).
-compile(export_all).

qsort([]) ->
    [];
qsort([X|Xs]) ->
    qsort([Y || Y <- Xs, Y<X]) ++
     [X] ++
     qsort([Y || Y <- Xs, Y>=X]).
```

Simplest just to export everything

werl/erl REPL



- Much like ghci

Test Data

- Create some test data; in foo.erl:

```
random_list(N) ->
    [random:uniform(1000000) || _ <- lists:seq(1,N)].
```

Side-effects!

Instead of [1..N]

- In the

```
L = foo:random_list(200000).
```

Timing calls



Benchmarking

Binding a name… c.f. **let**

Macro: current module name

```
benchmark(Fun,L) ->
    Runs = [timer:tc(?MODULE,Fun,[L])
            || _ <- lists:seq(1,100)],
    lists:sum([T || {T,_} <- Runs]) /
        (1000*length(Runs)).
```

- 100 runs, average & convert to ms

```
80> foo:benchmark(qsort,L).
330.88
```

Parallelism

```
34> erlang:system_info(schedulers).
4
```



Parallelism in Erlang

- Processes are created *explicitly*

```
Pid = spawn_link(fun() -> …Body… end)
```

- Start a process which executes …Body…
- **`fun() -> Body end`** ~ **`\() -> Body`**
- **`Pid`** is the *process identifier*

Parallel Sorting

```
psort([]) ->
    [];
psort([X|Xs]) ->
    spawn_link(
      fun() ->
        psort([Y || Y <- Xs, Y >= X])
      end),
    psort([Y || Y <- Xs, Y < X]) ++
          [X] ++
          ???.
```

Sort second half in parallel…

But how do we get the result?

Message Passing

Pid ! Msg

- Send a message to Pid
- *Asynchronous*—do not wait for delivery

Message Receipt

```
receive
    Msg -> …
end
```

- Wait for a message, then bind it to Msg

Parallel Sorting

```
psort([]) ->
    [];
psort([X|Xs]) ->
    Parent = self(),
    spawn_link(
      fun() ->
        Parent !
            psort([Y || Y <- Xs, Y >= X])
      end),
    psort([Y || Y <- Xs, Y < X]) ++
          [X] ++
          receive Ys -> Ys end.
```

The Pid of the executing process

Send the result back to the parent

Wait for the result *after* sorting the first half

Benchmarks

```
84> foo:benchmark(qsort,L).
327.13
85> foo:benchmark(psort,L).
474.43
```

- Parallel sort is slower! *Why?*

Controlling Granularity

```
psort2(Xs) -> psort2(5,Xs).

psort2(0,Xs) -> qsort(Xs);
psort2(_,[]) -> [];
psort2(D,[X|Xs]) ->
    Parent = self(),
    spawn_link(fun() ->
      Parent !
        psort2(D-1,[Y || Y <- Xs, Y >= X])
       end),
    psort2(D-1,[Y || Y <- Xs, Y < X]) ++
    [X] ++
    receive Ys -> Ys end.
```

Benchmarks

```
84> foo:benchmark(qsort,L).
327.13
85> foo:benchmark(psort,L).
474.43
86>
foo:benchmark(psort2,L).
165.22
```

- 2x speedup on 2 cores (x2 hyperthreads)

Profiling Parallelism with Percept

File to store profiling information in

{Module,Function, Args}

```
87> percept:profile("test.dat",{foo,psort2,[L]},[procs]).
Starting profiling.
ok
```

Profiling Parallelism with Percept

Analyse the file, buliding a RAM database

```
88> percept:analyze("test.dat").
Parsing: "test.dat"
Consolidating...
Parsed 160 entries in 0.078 s.
    32 created processes.
    0 opened ports.
ok
```

Profiling Parallelism with Percept

Start a web server to display the profile on this port

```
90> percept:start_webserver(8080).
{started,"JohnsTablet2012",8080}
```

Profiling Parallelism with Percept



Profiling Parallelism with Percept



Select	Pid	Lifetime	Entrypoint	Name	Processes Parent
	<0.756.1>		undefined	undefined	undefined
	<0.5834.667>		foo:'-psort2/2-fun-0-'/0	undefined	<0.756.1>
	<0.5835.667>		foo:'-psort2/2-fun-0-'/0	undefined	<0.5834.667>
	<0.5836.667>		foo:'-psort2/2-fun-0-'/0	undefined	<0.756.1>
	<0.5837.667>		foo:'-psort2/2-fun-0-'/0	undefined	<0.5836.667>
	<0.5838.667>		foo:'-psort2/2-fun-0-'/0	undefined	<0.756.1>
	<0.5839.667>		foo:'-psort2/2-fun-0-'/0	undefined	<0.5837.667>
	<0.5840.667>		foo:'-psort2/2-fun-0-'/0	undefined	<0.5836.667>
	<0.5841.667>		foo:'-psort2/2-fun-0-'/0	undefined	<0.5837.667>
	<0.5842.667>		foo:'-psort2/2-fun-0-'/0	undefined	<0.5839.667>
	<0.5843.667>		foo:'-psort2/2-fun-0-'/0	undefined	<0.756.1>
	<0.5844.667>		foo:'-psort2/2-fun-0-'/0	undefined	<0.756.1>
	<0.5845.667>		foo:'-psort2/2-fun-0-'/0	undefined	<0.5836.667>
	<0.5846.667>		foo:'-psort2/2-fun-0-'/0	undefined	<0.5843.667>
	<0.5847.667>		foo:'-psort2/2-fun-0-'/0	undefined	<0.5840.667>
	<0.5848.667>		foo:'-psort2/2-fun-0-'/0	undefined	<0.5838.667>
	<0.5849.667>		foo:'-psort2/2-fun-0-'/0	undefined	<0.5838.667>
	<0.5850.667>		foo:'-psort2/2-fun-0-'/0	undefined	<0.5848.667>
	<0.5851.667>		foo:'-psort2/2-fun-0-'/0	undefined	<0.5834.667>
	<0.5852.667>		foo:'-psort2/2-fun-0-'/0	undefined	<0.5834.667>
	<0.5853.667>		foo:'-psort2/2-fun-0-'/0	undefined	<0.5851.667>

Correctness

```
91> foo:psort2(L) == foo:qsort(L).
false
92> foo:psort2("hello world").
" edhllloorw"
```

Oops!

What's going on?

```
psort2(D,[X|Xs]) ->
    Parent = self(),
    spawn_link(fun() ->
      Parent ! …
       end),
    psort2(D-1,[Y || Y <- Xs, Y < X]) ++
    [X] ++
    receive Ys -> Ys end.
```

What's going on?



Message Passing Guarantees

Message Passing Guarantees



Tagging Messages Uniquely

```
Ref = make_ref()
```

- Create a globally unique reference

```
Parent ! {Ref,Msg}
```

- Send the message tagged with the reference

```
receive {Ref,Msg} -> … end
```

- Match the reference on receipt… picks the right message from the mailbox

A correct parallel sort

```
psort3(Xs) ->
    psort3(5,Xs).

psort3(0,Xs) ->
    qsort(Xs);
psort3(_,[]) ->
    [];
psort3(D,[X|Xs]) ->
    Parent = self(),
    Ref = make_ref(),
    spawn_link(fun() ->
      Parent ! {Ref,psort3(D-1,[Y || Y <- Xs, Y >= X])}
    end),
    psort3(D-1,[Y || Y <- Xs, Y < X]) ++
      [X] ++
      receive {Ref,Greater} -> Greater end.
```

Tests

```
23> foo:benchmark(qsort,L).
329.48
24> foo:benchmark(psort3,L).
166.66
25> foo:qsort(L) == foo:psort3(L).
true
```

- Still a 2x speedup, and now it works ☺

Parallelism in Erlang vs Haskell

- Haskell processes *share memory*



Parallelism in Erlang vs Haskell

- Erlang processes each have their own heap



Pid ! Msg



In Haskell, *forcing to nf* is linear time

- Messages have to be *copied*
- No global garbage collection—each process collects its own heap

What's copied here?

```
psort3(D,[X|Xs]) ->
    Parent = self(),
    Ref = make_ref(),
    spawn_link(fun() ->
      Parent !
        {Ref,
          psort3(D-1,[Y || Y <- Xs, Y >= X])}
    end),
```

- Is it sensible to copy *all of Xs* to the new process?

Better

A small improvement—but Erlang lets us *reason* about copying

```
psort4(D,[X|Xs]) ->
    Parent = self(),
    Ref = make_ref(),
    Grtr = [Y || Y <- Xs, Y >= X],
    spawn_link(fun() ->
       Parent ! {Ref,psort4(D-1,Grtr)}
    end),
```

```
31> foo:benchmark(psort3,L).
166.3
32> foo:benchmark(psort4,L).
152.6
```

Benchmarks on 4 core i7

```
17> foo:benchmark(qsort,L).
414.07
18> foo:benchmark(psort4,L).
144.8
```

- Speedup: 2.9x on 4 cores/8 threads
  - (increased depth to 8)

Haskell vs Erlang

- Sorting (different) random lists of 200K integers, on 2-core i7

	Haskell	Erlang
Sequential sort	353 ms	312 ms
Depth 5 //el sort	250 ms	153 ms

- *Despite* Erlang running on a VM!

Erlang scales much better

Erlang Distribution

- Erlang processes can run on *different machines* with the same semantics
- No shared memory between processes!
- Just a little slower to communicate…

Named Nodes

`werl -sname baz`

- Start a node with a *name*

```
(baz@JohnsTablet2012)1> node().
baz@JohnsTablet2012

(baz@JohnsTablet2012)2> nodes().
[]
```

Node name is an atom

List of connected nodes

Connecting to another node

`net_adm:ping(Node).`

```
3> net_adm:ping(foo@JohnsTablet2012).
pong
```

Success—pang means connection failed

```
4> nodes().
[foo@JohnsTablet2012,baz@HALL]
```

Now connected to foo and other nodes foo knows of

Node connections



Gotcha! the Magic Cookie

- All communicating nodes must share the same *magic cookie* (an atom)
- Must be the same on all machines
  - By default, randomly generated on each machine
- Put it in $HOME/.erlang.cookie
  - E.g. cookie

A Distributed Sort

```
dsort([]) ->
    [];
dsort([X|Xs]) ->
    Parent = self(),
    Ref = make_ref(),
    Grtr = [Y || Y <- Xs, Y >= X],
    spawn_link(foo@JohnsTablet2012,
       fun() ->
             Parent ! {Ref,psort4(Grtr)}
       end),
    psort4([Y || Y <- Xs, Y < X]) ++
     [X] ++
     receive {Ref,Greater} -> Greater
end.
```

Benchmarks

```
5> foo:benchmark(psort4,L).
159.9
6> foo:benchmark(dsort,L).
182.13
```

- Distributed sort is *slower*
  - Communicating between nodes is slower
  - Nodes on the same machine are sharing the cores anyway!

OK…

An older slower machine… 2 cores no hyperthreading… silly to send it half the work

```
dsort2([X|Xs]) ->
    …
    spawn_link(baz@HALL,
       fun() ->
            ….
```

```
5> foo:benchmark(psort4,L).
159.9
6> foo:benchmark(dsort,L).
182.13
7> foo:benchmark(dsort2,L).
423.55
```

Distribution Strategy

- Divide the work into 32 chunks on the master node
- Send *one chunk at a time* to each node for sorting
  - Slow nodes will get fewer chunks
- Use the fast parallel sort on each node

Node Pool

- We need a pool of *available nodes*

```
pool() ->
    Nodes = [node()|nodes()],
    spawn_link(fun() ->
       pool(Nodes)
    end).
```

- We create a process to manage the pool, initially containing all the nodes

Node Pool Protocol



Node Pool Behaviour

```
pool([]) ->
   receive
      {available,Node} ->
         pool([Node])
   end;
pool([Node|Nodes]) ->
   receive
      {get_node,Pid} ->
         Pid ! {use_node,Node},
         pool(Nodes)
   end.
```

If the pool is empty, wait for a node to become

If nodes are available, wait for a request and give one out

Selective receive is really useful!

dwsort

```
dwsort(Xs) -> dwsort(pool(),5,Xs).

dwsort(_,_,[]) -> [];
dwsort(Pool,D,[X|Xs]) when D > 0 ->
    Grtr = [Y || Y <- Xs, Y >= X],
    Ref = make_ref(),
    Parent = self(),
    spawn_link(fun() ->
      Parent ! {Ref,dwsort(Pool,D-1,Grtr)}
    end),
    dwsort(Pool,D-1,[Y || Y <- Xs, Y < X]) ++
      [X] ++
      receive {Ref,Greater} -> Greater end;
```

Parallel recursion to depth 5

dwsort

```
dwsort(Pool,0,Xs) ->
    Pool ! {get_node,self()},
    receive
      {use_node,Node} ->
         Ref = make_ref(),
         Parent = self(),
         spawn_link(Node, fun() ->
            Ys = psort4(Xs),
            Pool ! {available,Node},
            Parent ! {Ref,Ys}
         end),
         receive {Ref,Ys} -> Ys end
    end.
```

A further optimisation: if we should use the *current* node, don't spawn a new process

Benchmarks

```
56> foo:benchmark(qsort,L).
321.01
57> foo:benchmark(psort4,L).
156.55
58> foo:benchmark(dsort2,L).
415.83
59> nodes().
[baz@HALL]
60> foo:benchmark(dwsort,L).
213.12
61> net_adm:ping('mary@CSE-360').
pong
62> nodes().
[baz@HALL,'mary@CSE-360']
63> foo:benchmark(dwsort,L).
269.71
```

Oh well!

- It’s quicker to *sort* a list, than to send it to another node and back!

Another Gotcha!

- All the nodes must be running *the same code*
  - Otherwise sending functions to other nodes cannot work
- We can *load the code* onto each node using remote procedure calls

rpc

```
rpc:call(Node,Module,FunName,Arguments)
```

- Calls Module:FunName(Arguments…) on the given Node

Call on *all* nodes

```
rpc_all(M,F,Args) ->
    [rpc:call(Node,M,F,Args)
     || Node <- [node()|nodes()]].
```

Load the code on all nodes

```
load() ->
    rpc_all(code,purge,[foo]),
    {ok,Code} = file:read_file("foo.beam"),
    rpc_all(file,write_file,
                  ["foo.beam",Code]),
    rpc_all(code,load_file,[foo]).
```

Write the compiled code on all nodes

Load the code we just uploaded

- Erlang distribution is for a *sec*
  network!

Summary

- Erlang parallelism is more explicit than in Haskell
- Processes do not share memory
- All communication is explicit by message passing
- Performance and scalability are strong points
- Distribution is easy
  - (But sorting is cheaper to do than to distribute☹)

References

- *Programming Erlang: Software for a Concurrent World*, Joe Armstrong, Pragmatic Bookshelf, 2007.



- *Learn you some Erlang for Great Good*, Frederic Trottier-Hebert , http://learnyousomeerlang.com/